

Laskuvarjohyppy

Heittoliikkeeseen kuuluu keskeisinä tekijöinä maan vetovoima sekä ilmanvastuksen aiheuttama nopeudelle vastakkaissuuntainen voima. Näitä voimia sovelletaan myös mallinnettaessa laskuvarjolla hyppäämistä.

Tarkastellaan ensin vapaata putoamista. Putoajaan vaikuttava voima aiheuttaa kiihtyvyyden, joka saa aikaan nopeuden. Yleisessä muodossa voimalle pätee

$$\overline{F} = m\overline{a},$$

missä m on tarkasteltavan kohteen massa ja $\overline{a}$ kiihtyvyys. Hyppääjään vaikuttavat voimat ovat siis painovoima $\overline{G} = m\overline{g}$, missä $\overline{g}$ on maan vetovoiman aiheuttama kiihtyvyys, sekä ilmanvastus, jota arvioidaan yhtälöllä

$$\overline{F} = -bv^2\hat{v} = -bv\overline{v}.$$

Tässä b on ilmanvastuskerroin, v skalaarinen nopeus, $\overline{v}$ nopeus vektorina ja $\hat{v}$ nopeuden suuntainen yksikkövektori. Liikkeyhtälö saa tällöin muodon

$$m \frac{\partial^2 \overline{r}}{\partial t^2} = m\overline{g} - b \left| \frac{\partial \overline{r}}{\partial t} \right| \frac{\partial \overline{r}}{\partial t},$$

missä on merkitty $\overline{a} = \frac{\partial^2 \overline{r}}{\partial t^2}$ ja $\overline{v} = \frac{\partial \overline{r}}{\partial t}$.

Tämä on heittoliikkeen differentiaaliyhtälö yleisessä muodossa.

Tehdään tarvittavat laskut laskuvarjohyppääjän lentoradan ja -nopeuden selvittämiseksi. Edellä esitetty ilmanvastusmalli on vain approksimaatio todelliselle ilmanvastukselle. On myös huomattava, että mikäli hyppääjä putoaa ilman varjoa, ilmanvastustekijä on aivan eri suuruinen kuin varjon kanssa.

Laskujen aluksi on syytä hävittää mahdollisista aiemmista laskuista jääneet muuttujat.

```
> restart;
```

Määritellään differentiaaliyhtälö normaaliryhmänä, jossa on erikseen x ja y-komponentit.

```
> ryhmä:= diff(x(t), t)=u(t), m*diff(u(t),  
t)=-b*sqrt(u(t)^2+v(t)^2)*u(t), diff(y(t), t)=v(t), m*diff(v(t),  
t)=-m*g-b*sqrt(u(t)^2+v(t)^2)*v(t);
```

$$\text{ryhma} := \frac{d}{dt} x(t) = u(t), m \left(\frac{d}{dt} u(t) \right) = -b \sqrt{u(t)^2 + v(t)^2} u(t), \frac{d}{dt} y(t) = v(t),$$

$$m \left(\frac{d}{dt} v(t) \right) = -m g - b \sqrt{u(t)^2 + v(t)^2} v(t)$$

Probleeman tuntemattomat funktiot ovat vaakasijaintia kuvaava $x(t)$ sekä korkeus $y(t)$ ja vastaavat nopeudet $u(t)$ ja $v(t)$.

```
> tuntemattomat := x(t), y(t), u(t), v(t);
      tuntemattomat := x(t), y(t), u(t), v(t)
```

Määritetään parametrit ja valitaan alkuehdot lentokoneesta hypättäessä. Oletetaan hitaasti lentävä koulutuslentokone.

```
> m:= 70: b:= 30: g:= 9.81:
> alkuehto:= x(0)=0, y(0)=1000, u(0)=20, v(0)=0;
      alkuehto := x(0) = 0, y(0) = 1000, u(0) = 20, v(0) = 0
```

Ratkaistaan differentiaaliyhtälöryhmä. Käytetään **startinit=true** -parametria, koska myöhemmin käytettävä **fsolve**-komento vaatii sitä.


```
> rtk:= dsolve({ryhma,alkuehto}, {tuntemattomat}, type=numeric,
      output=listprocedure, startinit=true);
      rtk := [t = (proc(t) ... end proc), u(t) = (proc(t) ... end proc),
      v(t) = (proc(t) ... end proc), x(t) = (proc(t) ... end proc), y(t) = (proc(t) ... end proc)]
```

Poimitaan ratkaisusta hyppääjän lentorataa ja nopeutta kuvaavat komponentit listaksi.

```
> rata:= subs(rtk, [x(t), y(t)]):
      nopeus:= subs(rtk, [u(t), v(t)]):
```

Piirretään kuva laskuvarjohyppääjän lentoradasta. Huomaa koordinaattiakselien eri skaalat!

```
> plot([rata[], 0..250], view=[0..10, 950..1000]);
```




```
> fsolve('rata[2](t)=0');
```

209.5020688

Lentorata ei muodostunut kovin pitkäksi ja laskeutuminen kilometrin korkeudelta tapahtui n. 3.5 minuutissa. Tarkastellaan vielä hyppääjän nopeutta.

```
> plot(nopeus, 0..250, view=[0..5, -6..20]);
```



```
> array([seq(nopeus(k), k=1..10)]);
```

1.39847293733887912	-4.27034825310804944
0.187609380336178743	-4.76782066528266580
0.0241774977076799381	-4.78395009633720392
0.00311122728553184019	-4.78434080128117501
0.000400330576333358652	-4.78434930795562608
0.0000515032279070221438	-4.78434948211397604
0.661337118742183835 10 ⁻⁵	-4.78434948550800243
0.840283652130947287 10 ⁻⁶	-4.78434948556210849
0.101321617622732455 10 ⁻⁶	-4.78434948556217332
0.817316838584510061 10 ⁻⁸	-4.78434948556245665

Vaakasuuntainen nopeus lähes häviää ja putoamisnopeus asettuu vajaaseen 4.8 m/s nopeuteen jo kahden ensimmäisen sekunnin aikana. Hyppääjän putoamisvauhti maahan on siis noin 17 km/h, joka on suurenpuoleinen jaloilla vastaanotettavaksi.

Tarkastellaan seuraavaksi tilannetta, jossa hyppääjä ei avaa varjoaan välittömästi. Tehdään vastaava tarkastelu kuin edellä, paitsi että ilmanvastuskerroin b muuttuu hyppääjän avatessa varjon. Oletetaan, että ilman varjoa hyppäävän ihmisen maksiminopeus (vaikuttavien voimien summa on 0) eli ns. terminaalinopeus on runsaat 200 km/h eli n. 60 m/s. Tällöin

ilmanvastuskerroin voidaan ratkaista yhtälöstä $F = m g - b v^2 = 0$, jolloin $b = \frac{m g}{v^2}$.

```
> b0:= 70*9.81/60^2;
```

$b_0 := 0.1907500000$

Jotta voimme muuttaa ilmanvastuskertoimen arvon $b_0 = 0.19075$ varjon auetessa arvoon $b = 30$, tarvitaan funktio, joka mallintaa muutoksen. Tällaiseksi sopii


```
> f := x -> 1 - exp(-exp(2*x - 0.5));
```

$$f := x \rightarrow 1 - e^{(-e^{(2x - 0.5)})}$$

Funktion derivaatta muuttuu nopeasti kohdassa yksi lähisällä. Tästä voi aiheutua ongelmia myöhemmin numeerisessa differentiaaliyhtölön ratkaisussa.

Funktion kuvaaja näyttää seuraavalta:

```
> plot(f(x), x=-10..10, view=[-10..10, -0.5..1.5]);
```



Määritetään putoamisyhtälöt uudestaan. Ilmanvastuskerroin määräytyy nyt funktiosta

```
> bb := t -> b0 + (b - b0)*f(t);
```

$$bb := t \rightarrow b_0 + (b - b_0) f(t)$$

ja uusi yhtälöryhmä on

```
> ryhmä2 := diff(x(t), t)=u(t), m*diff(u(t),  
t)=-bb(t-20)*sqrt(u(t)^2+v(t)^2)*u(t), diff(y(t), t)=v(t),
```

```
m*diff(v(t), t)=-m*g-bb(t-20)*sqrt(u(t)^2+v(t)^2)*v(t);
```

```
ryhma2 :=  $\frac{d}{dt} x(t) = u(t),$ 
```

```
70  $\left(\frac{d}{dt} u(t)\right) = -(30.00000000 - 29.80925000 e^{(-e^{(2t-40.5)})}) \sqrt{u(t)^2 + v(t)^2} u(t),$ 
```

```
 $\frac{d}{dt} y(t) = v(t),$ 
```

```
70  $\left(\frac{d}{dt} v(t)\right) = -686.70 - (30.00000000 - 29.80925000 e^{(-e^{(2t-40.5)})}) \sqrt{u(t)^2 + v(t)^2} v(t)$ 
```

Varjon on oletettu aukeavan 20 sekuntia hyppämishetken jälkeen.

Ratkaistaan yhtälöryhmä numeerisesti. Kyseessä oleva yhtälöryhmä on hankala ratkaistava numeerisesti ja oletusarvoilla `dsolve`-komento suoranaisesti laskee väärin. `Startinit=true` -määreellä laskenta perustuu aina annettuihin alkuarvoihin. Normaalisti laskenta perustuu jo laskettuihin välituloksiin. `Startinit=true` siis hidastaa laskentaa melkoisesti, koska laskenta aloitetaan aina alkuarvoista, mutta tulos on kuitenkin tarkempi.

```
> rtk2:= dsolve({ryhma2,alkuehto}, {tuntemattomat},  
output=listprocedure, type=numeric, startinit=true);
```

```
rtk2 := [t = (proc(t) ... end proc), u(t) = (proc(t) ... end proc),
```


```
v(t) = (proc(t) ... end proc), x(t) = (proc(t) ... end proc), y(t) = (proc(t) ... end proc)]
```

Poimitaan ratkaisusta hyppääjän lentorataa ja nopeutta kuvaavat komponentit listaksi.

```
> rata2:= subs(rtk2, [x(t), y(t)]):  
nopeus2:= subs(rtk2, [u(t), v(t)]):
```

Piirretään kuvaaja hyppääjän lentoradasta, määritetään alastulohetki ja alastulonopeus:

```
> plot([rata2[], 0..50], view=[0..200, 0..1000]);
```



```
> alas:=fsolve('rata2[2](t)=0',t,0..50);
```

```
      alas := 44.98260040
```

```
> nopeus2(alas);
```


```
[0.862895471794922922 10-20, -4.78435027443042405]
```

Tarkastellaan vielä putoamisnopeuksia kokonaistilanteen hahmottamiseksi. Putoamisnopeudet ajan funktiona ovat

```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

```
> display(plot(nopeus2, 0..50, view=[0..50, -60..20]), plot([[17,
y, y=-60..20], [20, y, y=-60..20]], color=black));
```



```
> rata2(17);
```

```
[ 163.552607016423138, 242.608600315603667 ]
```

Putoaminen tapahtuu varsin eri tavalla kuin avoimella varjolla. Vaikka varjon avautumishetkeksi asetettiin 20 s, avautuminen on alkanut askelfunktion mukaisesti muuttamaan ilmanvastuskerrointa radikaalisti ja siten jarruttamaan putoamista jo aikaisemmin. Nopeus alkaa vähetä noin 250 metrin korkeudessa, kun $t = 17$ s.

Piirretään kuvaaja kiihtyvyydelle. Koska ratkaisuproseduuria ei voi derivoida, lasketaan derivaatalle arvioita käyttämällä erotusosamäärää.

Poimitaan aluksi ratkaisuproseduurista nopeuden pystysuuntainen komponentti:

```
> numv:= subs(rtk2, v(t));
```

```
numv := proc(t) ... end proc
```

Muodostetaan aikavälistä 0 - 50 lista 0.5 sekuntin välein.


```
> aika:= seq(t/2, t=0..100):
```

Lasketaan nopeuden erotusosamäärä aika-listan pisteissä.

```
> derivpisteet:= seq((numv(t/2+0.00001)-numv(t/2))/0.00001,
t=0..100):
```


Piirretään kuvaaja derivaatasta ajan funktiona käyttäen edellä laskettuja erotusosamääriä.

```
> display(plot(zip((x, y)->[x, y], [aika], [derivpisteet])),  
plot([19, t, t=-10..30], color=black));
```



```
> (numv(19+0.00001)-numv(19))/0.00001;  
28.038
```

```
>
```

Voimakkain jarrutuskiihtyvyys (n. 28 m/s^2) tapahtui kohdassa $t = 19 \text{ s}$. Putoamisen kokonaisajasta avatun varjon kanssa kuluu yhä yli puolet, vaikka matka on noin viidennes.

Tehtävä

Milloin hyppääjän on viimeistään vedettävä varjon avausnarusta, kun oletetaan, että tämän on tapahduttava 3 sekuntia ennen askelfunktiossa käytettyä aikaa? Suurin mahdollinen nopeus maahan pudotessa ilman loukkaantumista on 10 m/s. Mikä on tällöin putoamisen kokonaisaika?

Linkit

tykillä ampuminen (tykki2.mws)
normaaliryhmä